http://www.mapleprimes.com/questions/145527-Is-This-Matrix-Primitive

Markiyan Hirnyk:

Let UL(2,Z) be the group of all the matrices of dimension 2 over the integers with the determinant equal to +/-1.
A matrix M in UL(2,Z) is called primitive if M is not equal to K^n for any K in UL(2,Z) and any positive integer n >= 2.
Is the matrix M:= Matrix([[27,5],[11,2]]) primitive? How to determine it in Maple?

First note, that a matrix is invertible, iff its determinant is invertible (over commutative rings), the group are the invertible matrices over Z. And primitive here means: there is no invertible K, such that M = K^n for 1 < n.

```
> restart; interface(version);
  with(LinearAlgebra):
```

Classic Worksheet Interface, Maple 17.00, Windows, Feb 21 2013, Build ID 813473

```
> M:=Matrix([[27,5],[11,2]]);
  'Determinant(M)': '%'=%;
```

$$M := \begin{bmatrix} 27 & 5 \\ 11 & 2 \end{bmatrix}$$

$$\text{Determinant}(M) = -1$$

If a power is invertible, then the group element is invertible. And since determinants and products of matrices do commute it follows: if there is K^n = M, then **n is odd** (for n even we would have Det(K)^(2*m) = 1, Det(K) = +-1) and **Det(K) = - 1**.

```
> 'Eigenvalues(M)';
  b:=min(%);
  a:=max(%%);
  ``;
  #'a<0'; is(%);
  'b=-1/a'; is(%);
```

$$\text{Eigenvalues}(M)$$

$$b := \frac{29}{2} - \frac{13\sqrt{5}}{2}$$

$$a := \frac{29}{2} + \frac{13\sqrt{5}}{2}$$

$$b = -\frac{1}{a}$$

true

The Jordan form of M it is a diagonal matrix with those eigenvalues (from which one can write down [all the] possible roots) and gives an idea, why M can not primitive.

Note that for integer powers the Eigenvectors remain the same, while the eigenvalues equal the same power of the respective eigenvalues, for $A . v = \lambda v$ we have

```
> #A.v= lambda*v;
  (A^2).v = A.(lambda*v);
  ``=lambda*(A.v);
  ``=lambda*(lambda*v), `etc.`;
```

$$(A^2) . v = A . (\lambda v)$$

$$= \lambda (A . v)$$

$$= \lambda^2 v, \text{etc.}$$

From that we know: if such a pair (K,n) would exist, then the Eigenvalues of K a n-th roots of those of M.

Eigenvalues are the roots of the char. polynomial and that has coefficients in the according ring Z, in this case even quite simple, $\text{CharacteristicPolynomial}(K, x) = x^2 - \text{Trace}(K)\, x + \text{Determinant}(K)$. If α, β are the roots, then we have $\alpha \beta = -1$ and $(\alpha + \beta) \in Z$

Now let be α some n-th root of a, so $\beta = -\frac{1}{\alpha}$, hence $\left(\alpha - \frac{1}{\alpha}\right) \in Z$ and $\alpha = x + y I$. Solve for y:

```
> 0 = 'evalc( Im( eval(alpha-1/alpha,alpha = x+y*I) ) )';
  y in {solve(%, y)};
```

$$0 = \text{evalc}\left(\Im\left(\left(\alpha - \frac{1}{\alpha}\right)\Big|_{\alpha = x + y I}\right)\right)$$

$$y \in \{0, \sqrt{-x^2 - 1}, -\sqrt{-x^2 - 1}\}$$

If y would be not zero then the 2 other cases remain.

```
> alpha:  % =eval(%, alpha=x+I*y);
  [eval(%, y=+sqrt(-x^2-1)), eval(%, y=-sqrt(-x^2-1))]; #expand(%);
  evalc(%);
```

$$\alpha = x + y\,I$$

$$[\alpha = x + \sqrt{-x^2-1}\,I, \alpha = x - \sqrt{-x^2-1}\,I]$$

$$[\alpha = x - \sqrt{x^2+1}, \alpha = x + \sqrt{x^2+1}]$$

But then α would be real. So y is zero in any case. Which says that the Eigenvalues are both real and thus are the unique, usual n-th roots of a and b.

Since the Eigenvectors of M form a basis of IR^2 the possible root K is unique (now over IR).

Let us see, how one re-writes transformations in terms of Eigenvalues and Eigenvectors (if they form a basis): for a given vector w one needs the coefficients in terms of the basis of Eigenvectors (with a choice of ordering) - this is done by forming the matrix consisting of Eigenvectors and after inverting one gets the coefficients by applying it to w, see below.

I do it 'manually' through $-\mathrm{Id}\,\lambda + (M\,.\,x)$, since Maple does not guarantee my desired

```
> 'M - a*IdentityMatrix(2)': NullSpace(%): op(1, %):
  u:= map(rationalize, %);

  'M - b*IdentityMatrix(2)': NullSpace(%): op(1, %):
  v:= map(rationalize, %);
  ``;
  'M.u=a*u'; map(expand, a*u): Equal(M.u, %);
  'M.v=b*v'; map(expand, b*v): Equal(M.v, %);
```

$$u := \begin{bmatrix} \frac{25}{22} + \frac{13\sqrt{5}}{22} \\ 1 \end{bmatrix}$$

$$v := \begin{bmatrix} \frac{25}{22} - \frac{13\sqrt{5}}{22} \\ 1 \end{bmatrix}$$

$$M\,.\,u = a\,u$$

true

$$M\,.\,v = b\,v$$

true

```
> Eig:='Matrix([u,v])';
  F:='Eig^(-1)';
  F:= map(expand, %):
  ``;
  w:=Vector(2, symbol=r):
  'M.w' = '(F.w)[1]*a*u + (F.w)[2]*b*v ';
  lhs(%) = map(expand, rhs(%)):
  %, 'w' = w;
  Equal(lhs(%%), rhs(%%));
  #'w' = w;
```

$$Eig := \mathrm{Matrix}([u, v])$$

$$F := \frac{1}{Eig}$$

$$M\,.\,w = a\,u\,(F\,.\,w)_1 + b\,v\,(F\,.\,w)_2$$

$$\begin{bmatrix} 27\,r_1 + 5\,r_2 \\ 11\,r_1 + 2\,r_2 \end{bmatrix} = \begin{bmatrix} 27\,r_1 + 5\,r_2 \\ 11\,r_1 + 2\,r_2 \end{bmatrix}, w = \begin{bmatrix} r_1 \\ r_2 \end{bmatrix}$$

true

This 'recipe' also works for the n-th root of M, since one knows the Eigenvectors and the Eigenvalues, it writes as

```
> 'K.w =(F.w)[1]*alpha*u + (F.w)[2]*(beta)*v', 'beta = -1/alpha';
```

$$K\,.\,w = u\,\alpha\,(F\,.\,w)_1 + v\,\beta\,(F\,.\,w)_2, \beta = -\frac{1}{\alpha}$$

Let us work through it for the special case n = 2*m + 1 = 7.

This is the only case, where the Eigenvalues add up to an integer (for that the the consideration at the end of this sheet). And thus the only case where M may have a root.

```
> alpha:=a^(1/7);

  'alpha - 1/alpha': '%'= simplify(%);
```

$$\alpha := \left(\frac{29}{2} + \frac{13\sqrt{5}}{2}\right)^{(1/7)}$$

$$\alpha - \frac{1}{\alpha} = 1$$

The representing Matrix in the 'standard' basis is formed by the columns of the images of the 'standard' base vectors

```
> 'K=Matrix([w1,w2])';

  w:=Vector([1,0]):
  (F.w)[1]*alpha*u + (F.w)[2]*(-1/alpha)*v:
  map(expand, %):
  map(radnormal, %):
  w1:=map(simplify, %,size);

  w:=Vector([0,1]):
  (F.w)[1]*alpha*u + (F.w)[2]*(-1/alpha)*v:
  map(expand, %):
  map(radnormal, %):
  w2:=map(simplify, %,size);
  ``;
  K:=Matrix([w1,w2]);
```

$$K = \mathrm{Matrix}([w1, w2])$$

$$w1 := \begin{bmatrix} \frac{19}{13} \\ \frac{11}{13} \end{bmatrix}$$

$$w2 := \begin{bmatrix} \frac{5}{13} \\ \frac{-6}{13} \end{bmatrix}$$

$$K := \begin{bmatrix} \frac{19}{13} & \frac{5}{13} \\ \frac{11}{13} & \frac{-6}{13} \end{bmatrix}$$

Then A is a 7-th root of M, but over the Rationals and not over Z (here we have used the 'standard' basis of Z^2).

```
> 'K^7': '%'=%;
```

$$K^7 = \begin{bmatrix} 27 & 5 \\ 11 & 2 \end{bmatrix}$$

We are done, if we can show, that the Eigenvalues do not add up to some integer - which they must, they give the linear coefficient of the characteristic polynomial.

```
> linTerm:='a^(1/(2*m+1))-a^(-1/(2*m+1))';
  ``=%;
  ``;
  [seq([j,eval(linTerm,m=j)], j=1 ..8)]: evalf(%): # the specific numerical values
  plot(%, style=point, symbolsize=20, color=red):
  plot(linTerm, m=1 .. 8, color = black):
  plots[display](%,%%);
  ``;
```

$$linTerm := a^{\left(\frac{1}{2\,m+1}\right)} - a^{\left(-\frac{1}{2\,m+1}\right)}$$

$$= \left(\frac{29}{2} + \frac{13\sqrt{5}}{2}\right)^{\left(\frac{1}{2\,m+1}\right)} - \left(\frac{29}{2} + \frac{13\sqrt{5}}{2}\right)^{\left(-\frac{1}{2\,m+1}\right)}$$




```
> 'diff(linTerm, m) < 0'; normal(%); is(%) assuming 1 < m;
```

$$\frac{\partial}{\partial m}\mathrm{linTerm} < 0$$

$$-\frac{2\ln\left(\frac{29}{2}+\frac{13\sqrt{5}}{2}\right)\left(\left(\frac{29}{2}+\frac{13\sqrt{5}}{2}\right)^{\left(\frac{1}{2m+1}\right)}+\left(\frac{29}{2}+\frac{13\sqrt{5}}{2}\right)^{\left(-\frac{1}{2m+1}\right)}\right)}{(2m+1)^2} < 0$$

true

Done.